
Appendix.  JAGS code for Bayesian analysis of silica data 

 

a.  JAGS code incorporating an animal prior with 10-fold uncertainty, untransformed 

cumulative exposure 

 

 model { 

  

  U ~ dlnorm(0, 1/1.175^2)    # 10-fold uncertainty as per NRC  

  beta_rat ~ dnorm(0.00296,1/.00063^2)    # rat prior for cumdose    

  beta <- beta_rat + log(U) 

    

  # ASSEMBLE CUMULATIVE EXPOSURE FROM JOB HISTORIES  

  # repeat cumulative exposure calculation for each of N subjects 

  for (i in 1:N) {      

    log(e[i]) <- beta*expos[i] # let e[i] be exp(ZB)  

    expos[i] <- sumex[i] 

    # calculate true exposure for each job task record (concentration * time) 

  }  

 

  # CONDITIONAL LIKELIHOOD FUNCTION 

  # Based on Method 2 from the Endo example in the WinBUGS manual 

  # I is number of risk sets and J[i] is number of subjects in risk set i 

  # CUMJ[i] is running total of J up to J[i-1] 

  for (i in 1:I) { 

    sumeset[i] <- sum(e[(CUMJ[i]+1):(CUMJ[i]+J[i])]) 

    for (j in 1:J[i]) { 

      p[i, j] <- e[CUMJ[i]+j] / sumeset[i] 

    } 

    case[(CUMJ[i]+1):(CUMJ[i]+J[i])] ~ dmulti( p[i,1:J[i]] , 1)   

  } 

}   

 

 

b.  JAGS code for analysis adjusting for exposure measurement error, untransformed 

cumulative exposure, and animal prior with 1-fold uncertainty 

 

model { 

  k <- 0.79756     # Berkson error multiplier 

  logsd <- sqrt(log(k*k+1)) # std dev in log exposure 

  tau <- 1/(logsd*logsd)  # expressed as precision 

  beta ~ dnorm(0.00296,1/.00063^2)    # rat prior for cumdose, U=1    

   

  # Measurement error model, applied to each record in the JEM file (d2)  

  for (i in 1:K){ 

    Terr[i] ~ dnorm(1,4)I(0.1,1.9)   # probability model for dust conversion error 

    Texp[i] <- jobexp[i] * Terr[i]   # true mean job task exposure 

  } 

    

  # ASSEMBLE CUMULATIVE EXPOSURE FROM JOB HISTORIES  

  # repeat cumulative exposure calculation for each of N subjects 

  for (i in 1:N) {      

    log(e[i]) <- beta*expos[i]  # let e[i] be exp(ZB)  

    expos[i] <- cumex[i] 

    # calculate true exposure for each job task record (concentration * time) 

    for (j in 1:nsteps[i]) { 

      expprod[i,j] <- whtime[i,j] * Iexp[i,j]   

      # MSMT ERROR FOR EACH PERSON-JOB-EXPOSURE ASSIGNMENT 

      logmean[i,j] <- log(Texp[whexprow[i,j]]/sqrt(1+k*k)+1.0E-8) 

      Ierr[i,j] ~ dnorm(0,tau)        # probability model for job task msmt error 

      Iexp[i,j] <- exp(logmean[i,j] + Ierr[i,j])   # true person job task exposure  

    } 

    cumex[i] <- sum(expprod[i,1:nsteps[i]]) # sum of exposures across job tasks 

    diff[i] <- cumex[i] - sumex[i] 

  }  

  meandiff <- mean(diff) 

  maxabsdiff <-max(abs(diff)) 

 

  # CONDITIONAL LIKELIHOOD FUNCTION 

  # Based on Method 2 from the Endo example in the WinBUGS manual 

  # I is number of risk sets and J[i] is number of subjects in risk set i 

  # CUMJ[i] is running total of J up to J[i-1] 

  for (i in 1:I) { 

    sumeset[i] <- sum(e[(CUMJ[i]+1):(CUMJ[i]+J[i])]) 

    for (j in 1:J[i]) { 

      p[i, j] <- e[CUMJ[i]+j] / sumeset[i] 

    } 

    case[(CUMJ[i]+1):(CUMJ[i]+J[i])] ~ dmulti( p[i,1:J[i]] , 1)   

  } 

}  

 


